




Tutorial

POWER BI

Formulación DAX



Javier mayoral Mendieta

Tabla de contenido

1. CALENDARAUTO():	2
2. FORMAT:	2
3. MEDIDAS:	3
4. FUNCIÓN SUM(NombreColumna) y SUMX(Tabla, Expresión)	3
4. CONDICIONAL IF(Test lógico, Valor si verdadero, valor si falso):.....	4
5. AND y OR:.....	4
6. CONTAR EN DAX:.....	5
7. CONCATENAR COLUMNAS Y TEXTOS.....	6
8. Función CALCULATE:	8
9. ACUMULADO:	9
10. COMPARATIVA AÑO ANTERIOR:	10
11. RANKING:.....	12
12. FUNCIÓN ALL:.....	13
13. FILTER:.....	13
14. EXPRESIÓN dt:	15
15. FUNCIÓN ADDCOLUMNS:	15
16. FUNCIÓN DATEDIFF:	16
17. SWITCH:	16
18. VAR:.....	17
19. SELECTEDVALUE (ColumnName, [AlternateResult])	20
20. TOOLTIPS:	28
21. REPRESENTAR DATOS POR SEMANAS	32
22. Conocer los clientes que no han comprado.	33
ANEXOS.....	39
TABLA DE FORMATOS:	39

1. CALENDARAUTO():

Nos crea una tabla con las fechas que incluyen nuestros datos. En el caso de que añadiéramos más fechas, la tabla se actualizaría automáticamente para incluir las nuevas fechas.

A continuación, podemos extraer de la columna fecha todos los datos de año, trimestre, semana, día semana, añadiendo siempre columnas:

```
Año = YEAR(CALENDARIO[Fecha_key])
```

Extrae el año de un campo con fecha.

También vale MONTH (mes), QUARTER (trimestre), WEEKNUM (día de la semana del año), WEEKDAY (día de la semana).

En la fórmula de WEEKDAY hay que indicar a Power BI cuál es el inicio de semana. Para indicar que el lunes es el primer día, se pone “,2” al final:

```
Día Semana = WEEKDAY(CALENDARIO[Fecha_key],2)
```

2. FORMAT:

Le aplica formato a cualquier medida o cálculo. Podemos utilizarlo mucho en fechas para extraer, por ejemplo, los días de la semana en letra:

```
Día Semana texto = FORMAT(CALENDARIO[Fecha_key], "dddd")
```

Transforma una fórmula de fechas de la semana en número, en días de la semana en letra. Si lo hicieramos con los meses sería igual, pero escribiendo: "mmmm".

Podemos también aplicar FORMAT a números para cambiar a moneda o decimales, etc.

3. MEDIDAS:

La realización de medidas es fundamental en Power BI. Realizaremos la cantidad de medidas, tanto implícitas, como explícitas, como sean necesarias. Siempre deberemos hacer más medidas explícitas, ya que éstas, agilizan mucho más el modelo a nivel de rendimiento, que añadir columnas nuevas. Además, hay que tener en cuenta algo muy importante: **las fórmulas que utilizamos para crear medidas explícitas, sólo se pueden realizar con medidas explícitas.**

Para organizarnos mejor, es muy apropiado crear una nueva tabla llamada “Medidas” para introducir dentro todas las medidas que vayamos realizando.

4. FUNCIÓN SUM(NombreColumna) y SUMX(Tabla, Expresión)

Su función es sumar una columna, campo o medida implícita. Por ejemplo, si tenemos una tabla en la que una columna o campo es “Unidades Vendidas” podemos sumarla para que nos arroje el valor del total de unidades vendidas:

Unidades vendidas = `SUM(VENTAS[Cantidad_vendida])`

Esta medida la podremos colocar en una tarjeta y nos arrojaría el valor de la cantidad de unidades vendidas.

Aunque existe un gran inconveniente. Cuando queremos sumar las cantidades para multiplicarlas por su precio de venta, esta función no sirve. Ya que no tiene sentido sumar los precios de venta.

Para solucionarlo podemos utilizar **SUMX(Tabla, Expresión)** y realizar el cálculo directamente sin tener que hacer 2 SUM de medidas implícitas y no tener que hacer nuevas columnas en una tabla.

La estructura de la fórmula nos pide que pongamos una tabla y la expresión que será el cálculo de multiplicar Cantidad x Precio. Pero lo hace línea a línea (de ahí la X al final de la fórmula) y de esta forma el resultado que nos arroja es la suma de todo el resultado, pero calculado fila a fila.:

Ingresos = `SUMX(VENTAS, VENTAS[Cantidad_vendida]*RELATED(PRODUCTOS[Precio]))`

Al no tener el campo de precio en la tabla de Ventas, aplicaremos la función **RELATED** para traernos el precio de la tabla de Productos. Sin añadir una nueva columna. Únicamente lo hacemos dentro de la medida.

4. CONDICIONAL IF(Test lógico, Valor si verdadero, valor si falso):

Funciona igual que en Excel. Es una fórmula condicional. Se utiliza mucho para poder cambiar o añadir expresiones dependiendo de un factor. Por ejemplo, si queremos añadir una columna en la que nos indique que, si las unidades vendidas en una venta son superiores a 1, que nos escriba “Venta múltiple” y caso contrario que nos arroje “Venta simple”, crearíamos la siguiente medida:

```
Tipo venta = IF(VENTAS[Cantidad_vendida]>1, "Venta múltiple", "Venta simple")
```

NOTA: Si queremos dejar el resultado en blanco, mejor que colocar "", es escribir la fórmula **BLANK()**.

Se puede hacer un IF anidado, como en Excel, para poner varias opciones, aunque para ello es mejor otra fórmula que veremos más adelante.

5. AND y OR:

Al igual que en Excel, se pueden introducir varias opciones dentro de la fórmula IF. Y podemos hacer que se cumplan todas (AND) o que se cumpla alguna (OR). Veamos dos ejemplos:

Queremos que, si Tipo Venta es igual a “Venta múltiple” (ejemplo anterior) y son de la marca “Truper”, nos escriba “Múltiple Truper” y en caso contrario que lo deje en blanco:

```
Ejemplo AND = IF(AND(VENTAS[Tipo venta]="Venta múltiple", RELATED(PRODUCTOS[Marca])="Truper"), "Múltiple Truper", BLANK())
```

Como se puede apreciar hemos añadido RELATED porque el campo de “Marca” no está en la tabla en la que estamos añadiendo la columna.

En el caso de que queramos que sea una cosa u otra, en vez de AND escribiríamos OR.

DAX ofrece una alternativa a AND y a OR, representando cada uno por los siguientes símbolos:

- AND → &&
- OR → ||

y así acortar la fórmula y que se pueda visualizar mejor. Por ejemplo, queremos que si productos son de la marca “Truper” y cuyo precio sea mayor de 300€, nos escriba “Producto estrella” y si no, que no ponga nada:

```
Truper >300€ = IF(PRODUCTOS[Precio]>300 && PRODUCTOS[Marca] = "Truper", "Producto estrella", BLANK())
```

Con && hacemos lo mismo que con AND. Y si hubiéramos colocado || sería una cosa o la otra.

Existe un truco en DAX cuando tenemos varios test lógicos y en vez de escribir todos ellos usando OR, podemos acortar usando `in{}`. Por ejemplo, queremos hacer una nueva columna en la que nos diga si es “fin de semana” o “entre semana”. Para ello en la tabla calendario insertaremos una nueva columna:

```
Fin semana/entre semana = IF(CALENDARIO[Día Semana] in {1,2,3,4,5}, "Entre semana", "Fin de semana")
```

Si queremos que el resultado sea que NO CONTENGA podemos añadir NOT:

```
Fin semana/entre semana = IF(NOT(CALENDARIO[Día Semana] in {1,2,3,4,5}), "Fin de semana", "Entre semana")
```

6. CONTAR EN DAX:

- COUNTA: Para cuando es texto.
- COUNT: Para cuando son números.
- COUNTROW: Contar filas (aunque estén algunas en blanco).
- COUNTBLANK: Contar filas en blanco
- DISTINCTCOUNT: Contar sin repetir

Las fórmulas son muy sencillas. COUNT y COUNTA cuentan registros que NO ESTÉN EN BLANCO en la columna a la que hacemos referencia en la fórmula. Si queremos que cuente todo, incluidos los blancos,

usaríamos COUNTROWS. Esta fórmula hace referencia a una tabla y no a un campo como las anteriores. Por ejemplo, si queremos contar las ventas que se han realizado:

```
Cantidad Ventas = COUNTROWS(VENTAS)
```

DISTINCTCOUNT cuenta una columna, pero sin repeticiones. Muy útil cuando, por ejemplo, queremos contar la cantidad de marcas que existen:

```
Cantidad de marcas = DISTINCTCOUNT(PRODUCTOS[Marca])
```

O por ejemplo la cantidad de productos vendidos:

```
Cantidad productos vendidos = DISTINCTCOUNT(VENTAS[Productos_Key])
```

7. CONCATENAR COLUMNAS Y TEXTOS

Podemos, al igual que en Excel, concatenar columnas y textos que insertemos personalmente con &.

Podemos realizar una medida o añadir una columna y aplicar el texto que queramos y unirlo a otra columna, medida o fórmula:

```
Concatenar = "Se han vendido: " & DISTINCTCOUNT(VENTAS[Productos_Key]) & " productos"
```

Y al colocarlo en una Tarjeta quedaría de esta forma:



Se han vendido:
2583 productos

Como vemos, la cantidad en número no tiene ningún formato. Y no podemos aplicar formato moneda a esta medida porque es texto. Existe una solución con la fórmula que ya vimos de FORMAT dentro de la medida:

Concatenar = "Se han vendido: " & FORMAT(DISTINCTCOUNT(VENTAS[Productos_Key]), "0,0") & " productos"

Y de esta forma nos saldrá el número con separación de miles sin decimales.

Se han vendido:
2.583 productos

Al final de este documento os adjunto unas tablas con los diferentes formatos.

También podemos unir 2 columnas añadiendo una columna nueva. En la tabla de PRODUCTOS podemos unir la columna Categoría, Marca y el Precio:

Categoría y Marca = PRODUCTOS[Categoría] & "-" & PRODUCTOS[Marca] & "-" & PRODUCTOS[Precio] & "€"

Productos_Key	Descripción	Costo	Precio	Marca	Categoría	Categoría y Marca	Truper > 300€
45091152	ESPATULA S JUANA ENDUIR 180 MM	450 €	562,5 €	Black and Decker	Herramienta	Herramienta-Black and Decker-562,5€	
45091155	ESCARDILLO CORAZON Y PALA 78 MM	166 €	207,5 €	Makita	Herramienta	Herramienta-Makita-207,5€	
413050090	PITON P/MAD SOLDADO 27 - 160 x 25	166 €	207,5 €	Truper	Herramienta	Herramienta-Truper-207,5€	
413050160	CLAVO ESCUADRA PUL 25 - 90	40 €	50 €	Karcher	Herramienta	Herramienta-Karcher-50€	
413050661	PITON 5/TOPE ESCUAD P/TACO 6 x 50	301 €	376,25 €	Bostich	Herramienta	Herramienta-Bostich-376,25€	
413050707	CLAVO CAJ PULIDO FRACC 12- 25	450 €	562,5 €	Husky	Herramienta	Herramienta-Husky-562,5€	
413051001	MENSULA DOBLE ALUM 570 MM	302 €	377,5 €	Raiker	Herramienta	Herramienta-Raiker-377,5€	
413051297	CINTA PELIGRO 8 CM x 200 MT	291 €	363,75 €	Efco	Herramienta	Herramienta-Efco-363,75€	
413051484	REGADERA CHAPA GALV N3. 12 LT	178 €	222,5 €	Irwin	Herramienta	Herramienta-Irwin-222,5€	
413051983	CLAVO CAJ PULIDO FRACC 12- 32	215 €	268,75 €	Ridgid	Herramienta	Herramienta-Ridgid-268,75€	
413052648	CLAVO C/CHATA 9-30	384 €	480 €	Bosch	Herramienta	Herramienta-Bosch-480€	
413062440	PITON P/MAD SOLDADO 25 - 110 x 25	123 €	153,75 €	Stihl	Herramienta	Herramienta-Stihl-153,75€	
413062561	MENSULA DOBLE ALUM 470 MM	446 €	557,5 €	Stanley	Herramienta	Herramienta-Stanley-557,5€	
413062573	MENSULA DOBLE ALUM 370 MM	73 €	91,25 €	Dewalt	Herramienta	Herramienta-Dewalt-91,25€	
413063495	MACHO URANGA A/ALEADO NF 7/16 x 20	286 €	357,5 €	Hoteche	Herramienta	Herramienta-Hoteche-357,5€	
422000590	MECHA MAX JUEGO A/R x 6 UNID	336 €	420 €	Prekul	Herramienta	Herramienta-Prekul-420€	
422000814	MACHO URANGA A/ALEADO NPT 1/2 x 14	381 €	476,25 €	Hyundai	Herramienta	Herramienta-Hyundai-476,25€	
422000824	MACHO URANGA A/ALEADO NPT 3/8 x 18	425 €	531,25 €	Herramientas Cuelvo	Herramienta	Herramienta-Herramientas Cuelvo-531,25€	
422000826	MACHO URANGA A/ALEADO NPT 3/4 x 14	324 €	405 €	Dremel	Herramienta	Herramienta-Dremel-405€	
422002167	CLAVO ESCUADRA PUL 19 - 50 x 100	225 €	281,25 €	Mikels	Herramienta	Herramienta-Mikels-281,25€	
422002998	MECHA MAX PLANA P/MAD 1/4	120 €	150 €	Skil	Herramienta	Herramienta-Skil-150€	
422003007	MAQ ZONTA PERF RTO-PALANCA PRESION	44 €	55 €	Urrea	Herramienta	Herramienta-Urrea-55€	
422003008	MECHA EZETA COLA CONICA 53.00 MM	282 €	352,5 €	Black and Decker	Herramienta	Herramienta-Black and Decker-352,5€	

8. Función CALCULATE:

=CALCULATE(Expression,[Filtrar1],...)

Calcula una totalización en base a filtros. Es decir, podemos calcular la suma de un campo, en base a un filtro. Por ejemplo, podemos calcular el total de Ingresos, pero sólo de año 2014:

Ingresos 2014 = CALCULATE([Ingresos], CALENDARIO[Año]=2014)

Vemos que, en esta fórmula, la primera parte, es la medida de Ingresos. Porque no se puede colocar un campo o columna. Debe ser una medida. Pero podríamos realizar el calculo dentro de la misma fórmula si no la tuviéramos:

Ingresos 2014 = CALCULATE(SUMX(VENTAS, VENTAS[Cantidad_vendida]* RELATED(PRODUCTOS[Precio])), CALENDARIO[Año]=2014)

Por supuesto podemos introducir dentro del filtro, expresiones como AND y OR. Por ejemplo, queremos los ingresos de los productos de la marca “Truper” y de la marca “Bosch”:

Ingresos Truper y Bosch = CALCULATE([Ingresos], PRODUCTOS[Marca] = "Truper" || PRODUCTOS[Marca] = "Bosch")

O incluso con in:

Ingresos Truper y Bosch = CALCULATE([Ingresos], PRODUCTOS[Marca] in {"Truper", "Bosch"})

Obviamente no podemos poner AND o && porque ningún producto tiene 2 marcas. Es una o la otra.

Con CALCULATE, si queremos añadir más filtros para se cumplan todos, no hace falta usar AND ni &&. Se coloca una coma y se añaden más. Por ejemplo, hacer la medida anterior pero además del año 2017:

Ingresos Truper y Bosch 2017 = CALCULATE([Ingresos], PRODUCTOS[Marca] in {"Truper", "Bosch"}, CALENDARIO[Año]=2014)

34.523.810 €

Ingresos Truper y Bosch

21.615.844 €

Ingresos Truper y Bosch 2017

9. ACUMULADO:

Podemos hallar el acumulado de una variable por años de dos formas muy sencillas:

- Con CALCULATE() y DATESYTD(CALENDARIO[FechaKey])
- Directamente con la fórmula TOTALYTD(expression, CALENDARIO [FechaKey]).

Acumulado de Ingresos 1 = `CALCULATE([Ingresos], DATESYTD(CALENDARIO[Fecha_key]))`


Acumulado Ingresos 2 = `TOTALYTD([Ingresos],CALENDARIO[Fecha_key])`

Cualquiera de las dos opciones es válida. Si en vez de acumular por años, lo queremos hacer por meses, utilizaríamos las fórmulas TOTALMTD para meses o TOTALQTD para trimestres. Y lo mismo para DATESYTD.

La mejor visualización que podemos hacer para el acumulado es una tabla simple con año, mes, la variable elegida (ingresos, costes, beneficio, etc.) y la medida de acumulado:

Año	Mes	Ingresos	Acumulado
2014	enero	57.836.575 €	57.836.575 €
2014	febrero	64.118.975 €	121.955.550 €
2014	marzo	72.100.650 €	194.056.200 €
2014	abril	68.019.575 €	262.075.775 €
2014	mayo	65.252.900 €	327.328.675 €
2014	junio	69.664.588 €	396.993.263 €
2014	julio	72.291.200 €	469.284.463 €
2014	agosto	61.468.800 €	530.753.263 €
2014	septiembre	66.929.575 €	597.682.838 €
2014	octubre	50.457.125 €	648.139.963 €
2014	noviembre	62.241.175 €	710.381.138 €
2014	diciembre	59.483.851 €	769.864.989 €
2015	enero	58.663.440 €	58.663.440 €
2015	febrero	59.086.349 €	117.749.789 €
2015	marzo	70.890.696 €	188.640.485 €
2015	abril	65.858.774 €	254.499.259 €
2015	mayo	68.477.530 €	322.976.789 €
2015	junio	64.982.376 €	387.959.165 €
Total		2.705.217.819 €	498.363.154 €

El gráfico más adecuado para representar el acumulado es el de líneas o de áreas apiladas. Siempre poniendo en eje X los meses, acumulado en el eje Y y en leyenda el año:



10. COMPARATIVA AÑO ANTERIOR:

Muy común en las empresas es comparar variables, como los ingresos, con el mismo periodo del año anterior. En DAX es muy fácil con la fórmula SAMEPERIODLASTYEAR(dates). Se insertará siempre en la fórmula de CALCULATE:

```
Comparativa = CALCULATE([Ingresos], SAMEPERIODLASTYEAR(CALENDARIO[Fecha_key]))
```

Después haremos una tabla simple en la que pondremos año, mes, ingresos (en este ejemplo), la comparativa y podemos además calcular una medida con la diferencia entre uno y otro:

Año	Mes_texto	Ingresos	Comparativa	Diferencia
2014	noviembre	62.241.175 €		
2014	diciembre	59.483.851 €		
2015	enero	58.663.440 €	57.836.575 €	826.865 €
2015	febrero	59.086.349 €	64.118.975 €	-5.032.626 €
2015	marzo	70.890.696 €	72.100.650 €	-1.209.954 €
2015	abril	65.858.774 €	68.019.575 €	-2.160.801 €
2015	mayo	68.477.530 €	65.252.900 €	3.224.630 €
2015	junio	64.982.376 €	69.664.588 €	-4.682.211 €
2015	julio	84.311.961 €	72.291.200 €	12.020.761 €
2015	agosto	56.494.969 €	61.468.800 €	-4.973.831 €
2015	septiembre	60.858.518 €	66.929.575 €	-6.071.058 €
2015	octubre	62.770.909 €	50.457.125 €	12.313.784 €
2015	noviembre	58.466.788 €	62.241.175 €	-3.774.388 €
2015	diciembre	56.914.853 €	59.483.851 €	-2.568.999 €
2016	enero	53.159.796 €	58.663.440 €	-5.503.644 €
2016	febrero	52.418.575 €	59.086.349 €	-6.667.774 €
Total		2.705.217.819 €	2.206.854.665 €	498.363.154 €

E incluso podemos hacer un formato condicional a la medida de “Diferencia” para que los negativos me los escriba en rojo. Para representar bien estos datos lo mejor es un gráfico combinado de líneas y barras para colocar en uno los Ingresos y en el otro la comparativa:



Como vemos, el primer año está eliminado ya que no tiene nunca con qué compararse. Por eso, en filtro del gráfico, hemos colocado sólo 2015, 2016 y 2017. En la tabla también podríamos hacerlo.

Si queremos personalizar la comparativa, por ejemplo meses, trimestres, días, etc. Podemos usar la fórmula de DATEADD(Dates, NumberOfIntervals, Intervalo). Por ejemplo, si lo hacemos sobre la última semana sería:

Comparativa última semana = `CALCULATE([Ingresos], DATEADD(CALENDARIO[Fecha_key], -7, DAY))`

El intervalo hay que ponerlo siempre en negativo.

11. RANKING:

Podemos hacer un ranking con cualquier medida sobre cualquier segmentación con la fórmula RANKX(ALLTabla), expresión). Por ejemplo, queremos calcular el ranking de ventas de los productos:

Ranking productos = `RANKX(ALL(PRODUCTOS), [Ingresos])`

En el caso de que tuviéramos una segmentación (cliente por ejemplo), el ranking, cuando llegue al producto que no se le ha vendido a este cliente, pasaría lo siguiente:

Descripción	Ingresos	Ranking productos
CURVA CHAPA GALV CORRUGADA 5 ANG 90	575 €	592
PASADOR PUERTA CHAPA 22x200 CINC	375 €	592
GRASERA MANO PICO HIDRAULICO 1 KG	344 €	594
MECHA EZETA CILINDRICA A/R 8.25 MM	289 €	595
ESCUADRA 5 JUANA MILIMETRADA 25 CM	288 €	596
CANILLA DUKE PVC ESF P/TAMBOR 3/4	275 €	597
CUCHARA GHERARDI SOLDADA N. 7	275 €	597
MECHA EZETA CILINDRICA A/R 10.50 MM	275 €	597
PASADOR PUERTA MADERA 22x200 BCEAD O	263 €	600
TUERCA ESTAMPADA CUADRADA 5/16	263 €	600
RIEL RTO-PERILLA PLAS C/PLOMO x 10	248 €	602
CAÑO VENTILACION CHAPA GALV 3	238 €	603
MAQ ZONTA RTO-PALANCA 42	213 €	604
ABRAZ A/INOXIDABLE 5504 9 - 13 MM		605
ABRAZ A/INOXIDABLE 5506 12 - 16 MM		605
ABRAZ A/INOXIDABLE 5508 13 - 19 MM		605
ABRAZ A/INOXIDABLE 5510 14 - 22 MM		605
ABRAZ AMERICANA 3000 14 - 20 MM		605
ABRAZ AMERICANA 3001 16 - 22 MM		605
ABRAZ AMERICANA 3001/5 20 - 28 MM		605
ABRAZ AMERICANA 3002 25 - 35 MM		605
ABRAZ AMERICANA 3002/5 30 - 40 MM		605
ABRAZ AMERICANA 3003 35 - 45 MM		605
ABRAZ AMERICANA 3003/5 38 - 50 MM		605
Total	40.056.726 €	1

En el momento que no tiene venta de los productos nos aparece en blanco los ingresos (normal si no hay venta) y el ranking se queda fijo en un número. Para que no pase esto, basta con poner un condicional con el test lógico de que en el caso de que ingresos esté vacío, que no haga nada y caso contrario que calcule el ranking:

```
Ranking productos = IF([Ingresos]=BLANK(), BLANK(), RANKX(ALL(PRODUCTOS), [Ingresos]))
```

Además, cuando aplicamos una segmentación el ranking que aparece es general y no sobre esa segmentación. Son dos cosas distintas. Para que nos aparezca el ranking individual sobre la selección, cambiaremos ALL por ALLSELECTED:

```
Ranking productos ind = IF([Ingresos]=BLANK(),BLANK(),RANKX(ALLSELECTED(PRODUCTOS), [Ingresos]))
```

Y así obtendremos un ranking para la selección.

12. FUNCIÓN ALL:

La función ALL() se utiliza para que cualquier medida no se altere con los filtros. Por ejemplo, si tengo la medida de Ingresos y aplico:

```
Ingresos sin filtros = CALCULATE(Ingresos, ALL(tabla Ventas))
```

tendremos la medida de los Ingresos, pero no le afectarán los filtros. Y ya que la tenemos, podemos hacer la división de total ventas (con algún filtro) entre el total ventas ALL para hallar el % que suponen esas ventas filtradas sobre el total de ventas. **Sobre todo, es bueno aplicar ALL cuando tenemos una tabla por categorías, por ejemplo, y no queremos que el % afecte y que siga haciendo alusión al total vendido y no por la propia categoría.**

13. FILTER:

FILTER(Tabla existente, filtro) crea una **NUEVA TABLA** pero filtrada de una ya existente. Por ejemplo, si queremos hacer una nueva tabla sólo con los productos Truper:

```
Productos Truper = FILTER(PRODUCTOS, PRODUCTOS[Marca] = "Truper")
```

Al crear una tabla no podemos añadir medida. Debemos añadir TABLA. De esta forma nos crea una tabla nueva basada en la tabla PRODUCTOS pero únicamente de los productos con la marca Truper. Así tenemos una nueva tabla con muchos menos registros y podemos trabajar con ella utilizando muchos menos recursos y haciendo que el rendimiento sea mayor.

Como esta fórmula es una tabla en sí misma, podemos insertarla dentro de cualquier fórmula que nos requiera una tabla como por ejemplo SUMX o insertarla donde nos requieran un filtro porque también tiene la función de filtro, como por ejemplo en CALCULATE. Si introducimos FILTER dentro de un SUMX, no crea la tabla físicamente. La crea virtualmente. Por eso no la veremos. Pero su función se ejecuta y reducimos la cantidad de operaciones que haría sin insertarla. Por ejemplo, puedo hallar los ingresos de los productos de la marca Truper:


```
Ingresaos Truper Filter = SUMX(FILTER(VENTAS, RELATED(PRODUCTOS[Marca]) = "Truper"),  
VENTAS[Cantidad_vendida]*RELATED(PRODUCTOS[Precio]))
```

Y si realizamos la siguiente medida:

```
Ingresaos Truper = CALCULATE([Ingresos], PRODUCTOS[Marca] = "Truper")
```

obtendríamos el mismo resultado.

FILTER lo podemos usar mucho en CALCULATE como he comentado. Sobre todo, cuando introducimos una medida dentro de la parte de filtro de CALCULATE. Power BI no deja hacerlo, dando un error:



Por esta razón, podemos insertar FILTER en vez de la medida y solucionar este problema:

```
Ingresaos Truper y Bosch con filter = CALCULATE([Ingresos],  
FILTER(PRODUCTOS, PRODUCTOS[Marca] in {"Truper", "Bosch"}))
```

14. EXPRESIÓN dt:

dt"fecha_formatoamericano"

Sirve para escoger una fecha en concreto. Por ejemplo, si queremos saber los ingresos entre 15-5-2017 y el 20-5-2017:


```
Ingresos dt = CALCULATE([Ingresos], CALENDARIO[Fecha_key] >= dt"2017-5-15",
    CALENDARIO[Fecha_key] <= dt"2017-5-20")
```

15. FUNCIÓN ADDCOLUMNS:

Esta función nos añade una tabla y a la vez podemos añadir las columnas que necesitemos. Su expresión es la siguiente:

Nueva Tabla = ADDCOLUMNS(Tabla, NombreColumna, Expresión,...)

Por ejemplo, podemos añadir una tabla CALENDARIO:



```
1 DATOS DE FECHAS =
2     ADDCOLUMNS(
3         CALENDARAU(),
4         "AÑO",YEAR([Date]),
5         "MES",MONTH([Date]),
6         "DIA",DAY([Date]),
7         "NOMBRE DIA",FORMAT([Date], "DDDD"),
8         "NOMBRE MES",FORMAT([Date], "MMMM"),
9         "NUMERO DIA",WEEKDAY([Date], 2),
10        "NUMERO SEMANA",WEEKNUM([Date], 2)
11    )
12 // Esta tabla nos genera los datos de fecha y extrae de manera separada el año, el mes y el dia
13
```

16. FUNCIÓN DATEDIFF:

La expresión es DATEDIFF(StartDate, EndDate, Interval).

Sirve para hallar la diferencia entre dos fechas. Por ejemplo, cuando queremos saber los días que dura una promoción o la edad de una persona. Interval es el resultado que queremos obtener en días, meses, etc.

Interval puede ser YEAR, MONTH, etc.

17. SWITCH:

La expresión de la fórmula es:

SWITCH(Expresión, Valor, Resultado, [Else])

Es parecido al condicional IF. Por ejemplo, cuando queremos calcular una nueva columna con varias funciones lógicas, si usáramos IF, deberíamos de anidar, resultando una fórmula muy extensa. Por esta razón podemos utilizar SWITCH de la siguiente manera:

```
Tipo de producto = SWITCH(TRUE(),
    PRODUCTOS[Precio]<=100, "Barato",
    PRODUCTOS[Precio]<=300, "Medio",
    PRODUCTOS[Precio]>300, "Caro",
    BLANK())
```

Aquí la función TRUE() se utiliza porque estamos comparando. Si en vez de comparar sólo quisiéramos cambiar una cosa por otra, no usaríamos TRUE():

```
Tipo de producto 2 = SWITCH('Productos Truper'[Tipo de producto],
    "Barato", "Producto barato",
    "Medio", "Producto medio",
    "Caro", "Producto caro",
    BLANK())
```

O por ejemplo cuando ya tenemos un ranking de ventas, podemos hacer una medida con SWITCH para otorgar una categoría según los puestos de ranking:

```
Tipo de Ranking = SWITCH(TRUE(),  
    [Ranking productos]<=10, "Top 10",  
    [Ranking productos]<=100, "Top 100",  
    BLANK())
```

También podemos usar SWITCH para que nos muestre íconos: UNICHAR(Número). Es un código de Google que se puede buscar en internet y para cada ícono nos da un valor que introducimos dentro de UNICHAR (escoger HTML code sólo el número). Por ejemplo, una estrella es el código 9733. Así podemos crear una medida:

Ícono estrella = UNICHAR(9733) → Estrella llena

Ícono estrella = UNICHAR(9734) → Estrella vacía

Podemos concatenar varios UNICHAR con & en la misma medida:

Ícono estrella = UNICHAR(9733) & UNICHAR(9733) & UNICHAR(9733) & UNICHAR(9733)

Esto nos coloca 4 estrellas llenas.

Ícono estrella = UNICHAR(9733) & UNICHAR(9733) & UNICHAR(9733) & UNICHAR(9733) & UNICHAR(9734)

Esto nos coloca 4 estrellas llenas y una vacía.

Como vemos es un poco tedioso escribir tantas veces lo mismo. Por esta razón podemos incluir la función de VAR.

18. VAR:

La función de “var” podemos usarla siempre dentro de una medida o nueva columna. Sirve para crear una variable TEMPORAL, ya que únicamente nos servirá para esa medida. Su sintaxis es:

var nombre_variable = expresión, campo o medida.

RETURN

Por ejemplo en el caso anterior podemos crear una variable para UNICHAR(9733) y otra para UNICHAR(9734) y de esta forma ya no volveremos a escribir todo el texto:

```
Ranking estrellas =  
    var s1=UNICHAR(9733)  
    var s2=UNICHAR(9734)  
    return  
  
    SWITCH(TRUE(),  
        [Ranking clientes]<=2, s1 & s1 & s1 & s1 & s1,  
        [Ranking clientes]<=4, s1 & s1 & s1 & s1 & s2,  
        [Ranking clientes]<=6, s1 & s1 & s1 & s2 & s2,  
        [Ranking clientes]<=9, s1 & s1 & s2 & s2 & s2,  
        s1 & s2 & s2 & s2 & s2)
```

Y de esta forma podemos añadir a la tabla de ranking de clientes, esta nueva medida:

Cliente	Ingresos	Ranking	Ranking estrellas
Ferretería Halcones	872.235.675 €	1	★★★★★
Ferretería González	505.003.989 €	2	★★★★★
Tlapalería El Encino	463.762.026 €	3	★★★★☆
Tienda Ferretera del Sur	431.147.326 €	4	★★★★☆
Tienda Orizaba	144.449.018 €	5	★★★★☆
La Real	73.708.191 €	6	★★★★☆
Tlapalería 3R	67.897.834 €	7	★★☆☆☆
Comercializadora Mario	59.531.453 €	8	★★☆☆☆
Casa Díaz	40.056.726 €	9	★★☆☆☆
Hermanos Ruiz	18.948.888 €	10	★☆☆☆☆
Comercios Lara	14.958.541 €	11	★☆☆☆☆
Casa Constructora América	13.518.153 €	12	★☆☆☆☆
Total	2.705.217.819 €	1	★★★★★

Existe un truco para no estar escribiendo `s1 & s1 & s1 & s1 & s1`: `REPT(valor, repeticiones)`.

Podemos escribir: `rept(s1, 5)` y nos pondrá `s1` cinco veces.

En vez de UNICHAR, podemos usar un EMOJI (por ejemplo ). Podemos ir a la una web de emojis y copiarlo y pegarlo dentro de la fórmula, sin el UNICHAR. Eso sí, deberá ir entre comillas.

Ranking estrellas =

```
var s1="⭐"  
var s2="☆"  
return  
  
SWITCH(TRUE(),  
[Ranking clientes]<=2, s1 & s1 & s1 & s1 & s1,  
[Ranking clientes]<=4, s1 & s1 & s1 & s1 & s2,  
[Ranking clientes]<=6, s1 & s1 & s1 & s2 & s2,  
[Ranking clientes]<=9, s1 & s1 & s2 & s2 & s2,  
s1 & s2 & s2 & s2 & s2)
```

y la nueva visualización nos quedaría:

Cliente	Ingresos	Ranking	Ranking estrellas emoji
Ferretería Halcones	872.235.675 €	1	⭐⭐⭐⭐⭐
Ferretería González	505.003.989 €	2	⭐⭐⭐⭐⭐
Tlapalería El Encino	463.762.026 €	3	⭐⭐⭐⭐☆
Tienda Ferretera del Sur	431.147.326 €	4	⭐⭐⭐⭐☆
Tienda Orizaba	144.449.018 €	5	⭐⭐⭐☆☆
La Real	73.708.191 €	6	⭐⭐⭐☆☆
Tlapalería 3R	67.897.834 €	7	⭐⭐☆☆☆
Comercializadora Mario	59.531.453 €	8	⭐⭐☆☆☆
Casa Díaz	40.056.726 €	9	⭐⭐☆☆☆
Hermanos Ruiz	18.948.888 €	10	⭐☆☆☆☆
Comercios Lara	14.958.541 €	11	⭐☆☆☆☆
Casa Constructora América	13.518.153 €	12	⭐☆☆☆☆
Total	2.705.217.819 €	1	⭐⭐⭐⭐⭐

Esta técnica de emojis la podemos usar siempre como si insertáramos texto. La podemos incluir en cualquier medida para adornar el resultado. Por ejemplo:

```
Ingresos con emoji = "Los ingresos son " & FORMAT([Ingresos], "0,0") & " 💰"
```

Los ingresos son
2.705.217.819 💰

Incluso se puede incluir el emoji dentro del formato de la visualización, dentro de título y dejar la medida normal:



19. SELECTEDVALUE (ColumnName, [AlternateResult])


Nos coloca el nombre de un valor de una columna de tabla seleccionada. Por ejemplo, tenemos una tabla con las Categorías y los ingresos:

Categoría	Ingresos
Abrasivos	15.115.340 €
Accesorios	22.830.838 €
Broqueros	566.741.780 €
Cadenas	7.570.890 €
Cerraduras	39.808.211 €
Cizallas	3.346.913 €
Discos	12.627.321 €
Herramienta	969.963.568 €
Herramienta de construccion	122.320.890 €
Herramienta inalambrica	15.293.314 €
Lijas	2.055.001 €
Maquinaria	796.522.576 €
Material	48.768.540 €
Motores	43.503.329 €
Pegamento	20.629.159 €
Pinturas	2.316.321 €
Pistolas	15.803.829 €
Total	2.705.217.819 €

Si hacemos la siguiente fórmula:

```
Selección = SELECTEDVALUE('CATEGORÍAS'[Categoría])
```

y la ponemos en una tarjeta, cuando seleccionemos una categoría rellenará con el nombre que hemos seleccionado:



Categoría	Ingresos
Abrasivos	15.115.340 €
Accesorios	22.830.838 €
Broqueros	566.741.780 €
Cadenas	7.570.890 €
Cerraduras	39.808.211 €
Cizallas	3.346.913 €
Discos	12.627.321 €
Herramienta	969.963.568 €
Herramienta de construccion	122.320.890 €
Herramienta inalambrica	15.293.314 €
Lijas	2.055.001 €
Maquinaria	796.522.576 €
Material	48.768.540 €
Motores	43.503.329 €
Pegamento	20.629.159 €
Pinturas	2.316.321 €
Pistolas	15.803.829 €
Total	2.705.217.819 €


Abrasivos
Selección

Pero si no hemos seleccionamos nada, sale BLANK(). Para que no pase esto, añadimos lo siguiente en la fórmula:


```
Selección = SELECTEDVALUE('CATEGORÍAS'[Categoría], "Seleccione una opción")
```

Esta función no sólo sirve para esto. Podemos hacer más, aunque sea más complejo. Veamos una de sus funciones con un ejemplo. Casi siempre en cualquier proyecto de Power BI tendremos los ingresos, los costes y el beneficio. Con SELECTEDVALUE podemos hacer una visualización en la que según la selección, nos aparezca la cantidad de uno de ellos:

19.1 Realizamos una nueva tabla con ESPECIFICAR DATOS



y nos aparecerá la siguiente ventana para hacer una nueva tabla:



Sólo haremos una columna que llamamos “Selección”. Y añadimos los 3 datos de Ingresos, Costes y Beneficio. A la tabla la llamamos “Seleccionador” (abajo a la izquierda). Nos aparecerá a la derecha con las demás tablas.

19.2 Hacemos un segmentador el campo de la nueva tabla en forma de mosaico:



19.3 Hacemos la siguiente medida con SWITCH:


```
Selección ICB =  
    var sel = SELECTEDVALUE(Seleccionador[Selección])  
    return  
    SWITCH(TRUE(),  
        sel="Ingresos", [Ingresos],  
        sel="Costes", [Costes],  
        sel="Beneficio", [Beneficio])
```

Llamamos a al valor seleccionado “sel” y con SWITCH le decimos que si está seleccionado “Ingresos” que me saque la medida de Ingresos y así con las demás.

19.4. Ahora podemos hacer varias visualizaciones. Por ejemplo, vamos a insertar un gráfico de evolución para Ingresos, Costes y Beneficio:

- Eje X: Ponemos como siempre año y mes.
- Eje Y: En este caso en vez de poner Ingresos o Gastos o Beneficio, añadimos la nueva medida “Selección ICB”.

De esta forma cada vez que, en el segmentador seleccionamos una de las 3 opciones, se reflejará en el gráfico, la medida correspondiente:



Si seleccionamos Ingresos, la gráfica mostrará la evolución de los Ingresos. Si seleccionamos Costes, la gráfica mostrará los Costes. E igual con Beneficio.


Pero, el título, no refleja lo que realmente está mostrando. Esto se puede cambiar de la siguiente forma:

Dentro de “dar formato al objeto visual” y dentro de “General” tenemos título. Podemos ver que el título lo genera de forma automática Power BI. Pero se puede cambiar y personalizar. Aunque, lo que queremos es que el título sea dinámico. Es decir, que cuando esté seleccionado Ingresos, me ponga “EVOLUCIÓN DE LOS INGRESOS” y si seleccionamos Beneficio, me ponga “EVOLUCIÓN DE LOS BENEFICIOS”. E igual con costes.

Para ello, haremos una nueva medida:


```
Título gráfico = "Evolución de los " & SELECTEDVALUE(Seleccionador[Selección])
```

Con esta medida, estamos haciendo que el título sea dinámico. Según seleccionemos, escribirá una cosa u otra. Ahora dentro de título, tenemos un “fx”:



Y en: *¿En qué campo debemos basar esto?*, desplegamos y seleccionamos nuestra medida de título.

Y de esta forma, hemos insertado una medida en título para que éste, sea dinámico y nos escriba dependiendo de lo que seleccionemos:



Otro ejercicio que podemos hacer con **SELECTEDVALUE** es ayudar a que la imagen del dashboard, mejore notablemente. Podemos añadir, según la selección, una imagen. Veamos los pasos a seguir:


Insertamos un segmentador por país. En nuestro caso nos salen 2. México y Estados Unidos. Y queremos que cuando seleccionemos uno, nos aparezca la bandera del propio país. Así que iremos a Google a buscar las 2 imágenes de banderas. Pero en este caso, no las descargaremos, sino que copiaremos su URL. Pero antes haremos la siguiente medida:

```
Banderas =
var sel=SELECTEDVALUE(DISTRIBUIDOR[Pais])
return

SWITCH(TRUE(),
    sel="Estados Unidos", "https://www.rotuvall.es/wp-content/uploads/2019/07/Bandera-Estados-
    Unidos-EEUU-ROTUVALL-W.jpg",
    sel="México",
    "https://upload.wikimedia.org/wikipedia/commons/thumb/f/fc/Flag_of_Mexico.svg/2560px-
    Flag_of_Mexico.svg.png")
```


Cuando seleccionemos “México”, pondremos entre comillas la dirección URL de la imagen que hemos buscado. E igual con “Estados Unidos”.

Pero Power BI, con lo que tiene, no es capaz de reflejar la imagen. Debemos descargar una nueva visualización desde el panel de visualizaciones, haciendo clic en los 3 puntos al final de las visualizaciones:



Y hacemos clic en “Obtener más objetos visuales”. En el cuadro que nos aparece tenemos que buscar: Image by cloud. Nos aparecerá una visualización y la haremos clic sobre ella y la instalaremos.


Nos aparecerá incluida en nuestras visualizaciones:



Una vez descargada, la insertamos en nuestro dashboard:



En la parte de añadir datos a la visualización, donde pone “Image url”, insertaremos nuestra medida de “Banderas”. Y como no tenemos nada seleccionado, saldrá un mensaje: “The image undefined could not be loaded”. No pasa nada. Simplemente no tenemos ningún país seleccionado. Si seleccionamos uno de ellos, nos aparecerá la bandera:



Para solucionar la frase molesta cuando no tenemos nada seleccionado, podemos buscar otra imagen, que represente lo que se va a seleccionar. Por ejemplo, yo he buscado una bola del mundo, para que cuando no haya nada seleccionado, me muestre esta imagen. Obviamente, nos tenemos que ir a nuestra medida de “Banderas” y modificarla. Ya sabemos que tenemos una opción de la fórmula SWITCH, que en el caso de que no se seleccione nada, podemos escribir una opción. En este caso, pondremos la imagen de la bola del mundo:

```

Banderas =
var sel=SELECTEDVALUE(DISTRIBUIDOR[Pais])
return

SWITCH(TRUE(),
    sel="Estados Unidos", "https://www.rotuvall.es/wp-content/uploads/2019/07/Bandera-Estados-
Unidos-EEUU-ROTVALL-W.jpg",
    sel="México",
    "https://upload.wikimedia.org/wikipedia/commons/thumb/f/fc/Flag_of_Mexico.svg/2560px-
Flag_of_Mexico.svg.png",
    "https://img.pixers.pics/pho_wat(s3:700/F0/62/15/06/75/700_F062150675_804298a622f7f097df88521
4961416d4.jpg,700,700,cms:2018/10/5bd1b6b8d04b8_220x50-watermark.png,over,480,650,jpg)/fotomurales-bola-
del-mundo-3d-2.jpg.jpg")

```




Como veis, existen multitud de aplicaciones con SELECTEDVALUE. Tanto para mostrar datos como para mejorar notablemente el diseño de nuestro dashboard.

20. TOOLTIPS:


Con esta herramienta, podemos hacer que, cuando pasemos el cursor por encima de un dato de una visualización, nos despliegue una tabla con datos, que nosotros podemos personalizar. Para ello tenemos que hacer lo siguiente:

20.1 Añadir una página nueva (que llamaremos Tooltip) y hacemos que esta página sea destinada a ser Tooltip: dentro de “Dar formato a la página de informe”, en “Información de la página”, hacemos clic en “Permitir el uso co....” :



En nuestro dashboard se abre un cuadro en el que podemos introducir cualquier medida o incluso visualización para que, cada vez que pasemos por encima de un dato, se active:





Podemos también cambiar el tamaño del tooltip como queramos en configuración del lienzo.

En nuestro ejemplo, he colocado los ingresos por país. Para que cuando vaya a la visualización que tenemos de Ingresos por Categoría, nos muestre los Ingresos, además, por país:

Ciudad	Ingresos
Bogotá	431.147.326 €
Buenos Aires	463.762.026 €
Ciudad de México	203.980.470 €
Córdoba	73.708.191 €
Cuzco	40.056.726 €
Lima	872.235.675 €
Medellín	18.948.888 €
Monterrey	67.897.834 €
Santiago	28.476.694 €
Tijuana	505.003.989 €
Total	2.705.217.819 €


The tooltip configuration ribbon on the right shows 'Añadir a lienzo vertical' (Add to vertical layout) set to 'Superior'.

Por supuesto podemos añadir lo que queramos. Por ejemplo, etiquetas con texto personalizado, imágenes, tarjetas para que nos indique más datos, etc.



Una vez hecho, nos iremos a la visualización y nos iremos a su formato. Dentro de General, hacemos clic en “Información sobre...”. En tipo elegimos Página de informe y debajo elegimos el nombre la página de tooltip que hemos creado.

De esta forma, cada vez que pasemos el ratón por encima de un valor de la tabla, se abre el tooltip creado:



Otro ejemplo de tooltip sería:

Análisis de la semana del 01/01/2007 al 07/01/2007

Total de Beneficios: \$4.823.047.563

Categoría	Ingresos Netos
Computers	\$3.220.093.041
Cameras and camcorders	\$2.571.556.824
TV and Video	\$1.365.288.845
Cell phones	\$895.184.333
Music, Movies and Audio Books	\$166.466.683
Audio	\$152.205.434
Total	\$8.370.795.161

% de devoluciones

0,88 %

Existe un truco muy bueno que es añadir al título de la tabla una medida (lo hemos visto antes) que nos represente algún otro dato. Por ejemplo, el % de devoluciones en vez de ponerlo como tarjeta.

Contoso en español

Archivo Inicio Insertar Modelado Ver Optimizar Ayuda Herramientas de tablas **Herramientas de medición**

Nombre: Título tooltip
Tabla inicial: Medidas

Formato: Texto
Categoría de datos: Sin clasificar

Nueva Medida medida rápida Cálculos

1 Título tooltip = "El porcentaje de devoluciones es de: " & FORMAT([% de devoluciones], "percent")

Análisis de la semana del 01/01/2007 al 07/01/2007

Total de Beneficios: \$4.823.047.563

El porcentaje de devoluciones es de: 0,88%

Categoría	Ingresos Netos
Computers	\$3.220.093.041
Cameras and camcorders	\$2.571.556.824
TV and Video	\$1.365.288.845
Cell phones	\$895.184.333
Music, Movies and Audio Books	\$166.466.683
Audio	\$152.205.434
Total	\$8.370.795.161

Incluso podemos añadir algún ícono para decirle que el % de devoluciones que pase de 1% que nos ponga una cruz y si no, una mano de correcto:

```

Título tooltip =
var icono = IF([% de devoluciones] < 0.01, "👉", "❌")
return
"El porcentaje de devoluciones es de: " & FORMAT([% de devoluciones], "percent") & " " & icono

```

21. REPRESENTAR DATOS POR SEMANAS

Podemos hacer una tabla matriz por Países por ejemplo y por días de la semana. El día de la semana lo pondremos en letra extrayéndolo de la tabla calendario.

Luego debemos de poner 2 tarjetas filtro. Uno de Año (Obligatorio para que no me sume todos los años de un lunes de la primera semana) y otra con número de semana DEL AÑO ya que lo vamos a hacer por semanas.

Además, queremos que me represente la fecha de inicio y la fecha de fin de la semana seleccionada. Para ello debemos ir a Query para añadir Fecha de inicio semana y fecha fin de la semana en la tabla Calendario:

The screenshot shows the Power BI Data Editor interface. On the left, the 'Calendario' table is selected in the 'Consultas [10]' pane. The 'Fecha' column is currently selected, as indicated by the green header bar. A context menu is open over the 'Fecha' column, specifically the 'Fecha' section of the 'Análizar' (Analyze) tab. The menu shows various time intelligence functions: 'Antigüedad', 'Solo fecha', 'Analizar', 'Año', 'Mes', 'Trimestre', 'Semana', 'Dia', 'Restar días', 'Combinar fecha y hora', 'Más antiguo', and 'Más reciente'. Below the menu, the 'Fact-Ventas' table is visible, showing columns for 'FechaKey', 'Año', 'Semestre', 'Mes', 'Trimestre', 'Semana del año', 'Semana del mes', 'Dia', 'Restar días', 'Combinar fecha y hora', 'Más antiguo', and 'Más reciente'. The 'Semana' and 'Dia' columns are currently expanded, showing their respective table structures. The main data grid for the 'Calendario' table shows dates from January 1, 2007, to February 10, 2007, with columns for 'FechaKey', 'Año', 'Semestre', 'Mes', 'Trimestre', 'Semana', 'Dia', 'Restar días', 'Combinar fecha y hora', 'Más antiguo', and 'Más reciente'.

Y ahora podemos realizar la siguiente medida para añadir un título que nos indique que fechas de la semana son de la que hemos elegido:

```
Título Fechas = "Análisis de la semana del " & FIRSTDATE('Dim-Calendario'[Inicio de la semana]) & " al "
& FIRSTDATE('Dim-Calendario'[Final de la semana])
```

Al ser una medida, no podemos incluir una columna o campo como ya sabemos. Por ello usamos el comando FIRSTDATE. De esta forma, nos quedará todo así:

País	lunes	martes	miércoles	jueves	viernes	sábado	domingo	Total
Russia	\$52.529	\$33.628	\$37.033	\$49.575	\$52.409	\$33.435	\$26.174	\$284.782
Singapore	\$19.851	\$38.360	\$52.890	\$17.901	\$22.685	\$31.611	\$39.928	\$223.226
Italy	\$18.987	\$32.847	\$39.385	\$10.000	\$33.982	\$17.687	\$64.418	\$217.305
Taiwan	\$25.700	\$24.015	\$25.232	\$29.242	\$53.900	\$31.426	\$10.469	\$199.984
Thailand	\$30.576	\$29.055	\$22.342	\$56.602	\$21.837	\$11.238	\$24.582	\$196.232
Armenia	\$22.113	\$47.896	\$10.151	\$25.497	\$44.265	\$12.805	\$14.714	\$177.441
Bhutan	\$23.418	\$23.361	\$15.977	\$17.268	\$25.363	\$18.864	\$36.132	\$160.383
Kyrgyzstan	\$27.677	\$20.074	\$23.382	\$22.333	\$16.989	\$5.096	\$41.069	\$156.620
South Korea	\$18.750	\$18.480	\$5.266	\$15.158	\$51.415	\$22.677	\$18.304	\$150.050
Greece	\$4.004	\$10.744	\$3.912	\$2.557	\$9.941	\$28.016	\$5.415	\$64.589
Poland	\$13.032	\$19.734	\$6.928	\$4.992	\$3.488	\$3.454	\$4.512	\$56.140
Slovenia	\$10.002	\$2.600	\$7.322	\$19.408	\$6.704	\$3.288	\$4.656	\$53.979
the Netherlands	\$11.071	\$10.432	\$1.120	\$6.624	\$3.172	\$2.781	\$17.020	\$52.220
Total	\$5.621.286	\$5.983.598	\$5.925.128	\$6.040.272	\$5.983.431	\$6.236.997	\$6.202.233	\$41.992.944

En los segmentadores de Año y Semana, debemos aplicar la opción de tener siempre algo seleccionado, para que no haya confusiones en la lectura de la tabla.

22. Conocer los clientes que no han comprado.

Este es un dato muy importante en cualquier empresa. Saber cuándo un cliente ha dejado de comprar. Lo único que debemos saber antes es la variable de tiempo. Es decir, necesitamos saber el periodo que la empresa considera que un cliente ya no compra (1 mes, 2 meses, 1 año, etc.).

Primero haremos una tabla con todos nuestros clientes. En segundo lugar, crearemos una nueva medida que nos indique el último día de compra. Esto lo podemos hacer con la fórmula de MAX:

```
Última fecha compra = MAX(data[InvoiceDate])
```

Para el intervalo de tiempo, es decir, la cantidad de meses, días o años que un cliente se puede considerar “Perdido”, podemos hacer dos cosas:

- Hacer la cantidad estática (fija).
- Hacer la cantidad dinámica (cambiante).

Si es estática es fácil. Pondremos el intervalo fijo en la fórmula. Vamos a hacer primero el estático:

```
Diferencia fechas = IF(  
    DATEDIFF(data[Última fecha compra], DATE(2011,12,09), MONTH) > 6,  
    "Perdido",  
    "Actual")
```

Como vemos en la medida, para la diferencia de fechas elegimos la fórmula de DATEDIFF. Y ponemos un condicional para que nos arroje un valor de “Perdido” o “Actual” dependiendo de si el intervalo de tiempo es mayor a 3 meses o no en este caso.

NOTA: En este ejemplo no hemos utilizado la fórmula de TODAY() que deberíamos de haberla utilizado, porque la última fecha de pedidos es el 9-12-11. Por esta razón hemos colocado DATE(2011,12,09)


Una vez hechas las dos medidas, las colocamos en una tabla:

Clientes actuales y perdidos

CustomerID	Última fecha compra	Diferencia fechas
18250	11/02/2011	Perdido
18256	20/12/2010	Perdido
18269	16/12/2010	Perdido
18280	07/03/2011	Perdido
12347	07/12/2011	Actual
12348	25/09/2011	Actual
12349	21/11/2011	Actual
12352	03/11/2011	Actual
Total	09/12/2011	Actual

Y así podemos ver en un intervalo de 6 meses los clientes que están perdidos o no.

Ahora si queremos utilizar un intervalo de meses dinámico debemos de añadir una parametrización. Esto se hace desde el menú de MODELADO añadiendo “Parámetro nuevo”:



Cuando hacemos clic, nos pregunta si queremos intervalo numérico o campos. En este caso elegimos numérico para indicarle el número de meses. Se nos abrirá la siguiente ventana:

Parámetros

X

Agregue parámetros a objetos visuales y expresiones DAX para que los usuarios puedan usar segmentaciones de datos para ajustar las entradas y ver resultados diferentes. [Más información](#)

¿Qué ajustará la variable?

Nombre

Tipo de datos

Mínimo

Máximo

Incremento

Valor predeterminado


Agregar segmentación en esta página

Crear


Cancelar

- Nombre: Nombre que le queremos dar.
- Tipo de datos: Podemos elegir número entero, decimal y decimal fijo.
- Mínimo: Elegimos el valor mínimo a mostrar.
- Máximo: Elegimos el valor máximo a mostrar.
- Incremento: Si queremos incrementar de uno en uno o más.

Y haremos clic en crear. Se nos creará la siguiente visualización:



Y además se nos crea una nueva tabla:



Con dos campos: “Intervalo” y “Valor de intervalo”.

Una vez hecho esto, nos dirigimos a la medida ya creada de “Diferencia de fechas” y donde hemos puesto el número de meses en número, lo cambiamos por “Valor de intervalo”:

```
Diferencia fechas = IF(
    DATEDIFF(data[Última fecha compra], DATE(2011,12,09), MONTH) > Intervalo[Valor de Intervalo],
    "Perdido",
    "Actual")
```

para que nos coja el valor que le hemos dado en la visualización. Si la modificamos, la medida también se modifica y hacemos dinámico el intervalo de tiempo.

Además, podemos añadir 2 medidas que nos indique el número de clientes perdidos y el número de clientes actuales con la fórmula de CALCULATE y FILTER:

```
Cantidad clientes perdidos = CALCULATE(DISTINCTCOUNT(data[CustomerID]), FILTER(data, [Diferencia fechas] = "Perdido"))
```

```
Cantidad clientes actuales = CALCULATE(DISTINCTCOUNT(data[CustomerID]), FILTER(data, [Diferencia fechas] = "Actual"))
```

2768

Cantidad clientes perdidos

3578

Cantidad clientes actuales

Y por último podemos hacer otra medida que nos calcule los meses que hace que un cliente no compra:

```
Meses sin compra = DATEDIFF([Última fecha compra], DATE(2011,12,09),MONTH)
```

Y lo añadimos a la tabla:

Clientes actuales y perdidos

CustomerID	Última fecha compra	Diferencia fechas	Meses sin compra
12346	18/01/2011	Perdido	11
12347	07/12/2011	Actual	0
12348	25/09/2011	Actual	3
12349	21/11/2011	Actual	1
12350	02/02/2011	Perdido	10
12352	03/11/2011	Actual	1
12353	19/05/2011	Perdido	7
12354	21/04/2011	Perdido	8
Total	09/12/2011	Actual	0

ANEXOS

TABLA DE FORMATOS:

```
1  -- FORMAT is a formatting function that formats a value based
2  -- on a format string.
3  -- Use a backslash (\) to display the next character
4  -- in the format string.
5  EVALUATE
6  {
7    ( "Percent",      FORMAT (                 0.742, "Percent" )      ),
8    ( "Currency (1)", FORMAT (                 1234.567, "$#,0.00" )      ),
9    ( "Currency (2)", FORMAT (                 1234.567, """US$$" "#,0.00" ) ),
10   ( "Date (1)",     FORMAT ( DATE ( 2019, 3, 28 ), "yyyy-mm-dd" )      ),
11   ( "Date (2)",     FORMAT ( DATE ( 2019, 3, 28 ), "m/d/yy" )      ),
12   ( "Date (Q)",    FORMAT ( DATE ( 2019, 3, 28 ), "\QQ yyyy" )      )
13 }
```

[TRY IT](#) [COPY](#)

#1 **DAX.do** ▾

Value1	Value2
Percent	74.20%
Currency (1)	\$1,234.57
Currency (2)	US\$ 1,234.57
Date (1)	2019-03-28
Date (2)	3/28/19
Date (Q)	Q1 2019

```
1  -- FORMAT accepts a set of predefined format strings to format numbers
2  EVALUATE
3  {
4    ( "General Number",  FORMAT( 12345.67, "General Number" )  ),
5    ( "Currency"       , FORMAT( 12345.67, "Currency"       )  ),
6    ( "Fixed"          , FORMAT( 12345.67, "Fixed"          )  ),
7    ( "Standard"       , FORMAT( 12345.67, "Standard"       )  ),
8    ( "Percent"        , FORMAT( 12345.67, "Percent"        )  ),
9    ( "Scientific"     , FORMAT( 12345.67, "Scientific"     )  ),
10   ( "True/False"     , FORMAT( TRUE, "True/False"     )  ),
11   ( "On/Off"          , FORMAT( FALSE, "On/Off"          )  ),
12   ( "Yes/No"          , FORMAT( TRUE, "Yes/No"          )  )
13 }
```